

SSSSSSSS	iiiiii	n n
S	i	nn n
SSSSSSS	i	n n n
S	i	n nn
SSSSSSSS	iiiiii	n n

-Sinclair---Information---Nouvelles-  
 No.1 août 1985 \$1.50

---

Rédacteur: Réal Gagnon ( 544 Hermine #2 Quebec P.Q. G1M 2G6 )  
 Ce bulletin est conçu avec l'aide du traitement de texte  
 TASWORD II de Tasman Software (version TS 2068)  
 et d'une imprimante SMITH-CORONA, Fastext 80.

---

\*\* SOMMAIRE \*\*

- 1.....ACTUALITE-SINCLAIR
- 2.....Les TRUCS du 2068
- 3.....Prog:R-Header (2068)
- 4.....Source d'information
- 5.....Logiciels:
  - WHITE LIGHTNING (Spectrum)
  - COMPASS (2068)
  - TASWIDE (2068)
- 6.....Matériel:
  - ROMSWITCH (2068)
- 7.....Graphisme
- 8.....Prochain numéro

SINCLAIR EN DIFFICULTE...

Tout ne va pas si bien pour Sinclair Research, ils auraient accumulé une dette de 10 millions de Livres. La production du QL est arrêtée pour permettre un écoulement des stocks. Une prévision de 200,000 QLs vendus avait été faite au début de l'année, mais seulement 60,000 ont été vendus jusqu'à maintenant. Rappelons que Sinclair détient 40 % du marché en Angleterre.

L'introduction du QL en Amérique se fait attendre. Selon le magazine TS HORIZONS, il est possible de se procurer des QLs américains par le catalogue d'AMERICAN EXPRESS, au prix de 499 \$ US.

NOUVEAUX SPECTRUMS...

Il y a certaines rumeurs qui circulent voulant que SINCLAIR soit sur le point de sortir un SPECTRUM de 128K. Un peu comme le TS 2068, un système de "BANK SWITCHING" serait incorporé. On parle également d'un SPECTRUM portable avec un écran plat et 2 microdrives incorporés.

BIZARRE...

Depuis plusieurs mois, de nombreuses compagnies vendent des livres et des logiciels pour le QL, alors qu'on sait même pas quand il sera disponible en Amérique. Sinclair affirme que 26,000 demandes de renseignement sur le QL lui sont parvenues à Boston.

NOUVEAU QL...

Alors que le QL n'est pas encore disponible ici, en Europe on parle d'un QL 2. Il aurait 256K en standard, au lieu des 128K actuels, les 4 logiciels fournis avec le QL seraient en ROM et non plus sur microdrive. Plus un châssis d'extension avec 3 connecteurs pour les périphériques.

TOUJOURS LE QL...

Voici une chose intéressante, en France, le QL existe en version française. Les logiciels fournis avec la machine ainsi que leurs manuels sont entièrement traduits. Mais attention, ne pensez pas acheter votre QL ici et trouvez un endroit en France pour se procurer les logiciels français car ceux-ci ne tournent que sur des QLs français. Il est assuré que si des QLs sont vendus ici, ce sera des QLs américains donc pas de logiciel français pour nous. A moins que...

2068 ALIVE...

Au Portugal, selon TS Horizons, TIMEX aurait sorti un TS 2068. Il possède les mêmes caractéristiques que le 2068 américain sauf que le connecteur arrière est identique à celui de SPECTRUM. Donc il peut utiliser les périphériques du SPECTRUM alors que nous, pas. Pas bête comme idée.

ORDI-5...

Le seul endroit à Québec où l'on peut trouver ORDI-5, magazine français pour Sinclair seulement, c'est la Bibliothèque Centrale.

---- LES TRUCS DU 2068 ----

-SCROLL...

Si nous tapons ce programme:

```
10 FOR i=0 TO 10000
20 PRINT i
30 NEXT i
```

le message scroll? fera son apparition après chaque écran plein.

Pour remédier à cette situation, nous pouvons agir sur une variable-système appelé SCR CT, cela signifie SCROLL COUNTS. Si nous tapons à la ligne 15:

```
15 POKE 23692,255
```

scroll? est maintenant absent.

Mais il existe une méthode plus simple, l'emploi de la commande INPUT.

```
10 FOR i=0 TO 10000
20 INPUT ""
30 PRINT i: NEXT i
```

-LOAD...

Les longs programmes sont habituellement enregistrés en plusieurs parties sur une cassette. En premier lieu, un programme, dit "LOADER", est chargé, puis un écran de présentation et enfin le programme principal. Mais ce n'est pas très esthétique quand le nom du programme vient ruiner notre écran de présentation.

Pour éviter cette situation, nous pouvons faire:

```
POKE 23570,16
dans le petit programme de
```

chargement et dans le programme principal, on doit faire:

```
POKE 23570,6
```

sinon les commandes PRINT et LIST ne fonctionneront pas très bien.

-INKEY\$...

Supposons que nous avons un programme avec une ligne comme ceci: 50 IF INKEY\$="Y" THEN  
GOTO 100

si le clavier est en mode minuscule et que vous pressez "Y", Le 2068 verra "y" et passera à la ligne suivante.

Voici quelques POKES qui mettent le clavier dans le mode désiré:

```
POKE 23658,8 Mode MAJUSCULE
POKE 23658,0 Mode MINUSCULE
POKE 23617,2 Mode GRAPHIQUE
POKE 23617,1 Mode EXTENDED
```

Pour démontrer l'effet de ces POKES, essayez:

```
10 POKE 23658,valeur
ou
POKE 23617,valeur
20 INPUT LINE a$
```

Il y a aussi une autre solution à ce problème, si on fait:

```
PEEK 23556
```

la valeur retournée sera le code ASCII de la touche qui est pressé à ce moment. Ces valeurs correspondent à la lettre qui se trouve en haut, à gauche, sur la touche (A à Z, 0 à 9).

-INPUT...

En BASIC Sinclair, la commande INPUT possède un gros défaut.

En effet, ENTER fait que l'information sur l'écran est effacée. Le petit truc qui suit va garder cette information sur l'écran.

Pour ce faire, nous allons employer une routine dans la ROM qui "scroll" l'écran de 1 ligne vers le haut. Veuillez noter l'apostrophe, c'est très important.

```
Ex #1: 10 INPUT "Mois:";m$'
        USR 2361,USR 2361
        20 INPUT "Annee:";a'
        USR 2361,USR 2361
```

```
Ex #2: 10 INPUT AT 0,0;"Mois:"
        ;m$'USR 2361
        20 INPUT AT 0,0;"Annee:"
        ";a'USR 2361
```

Il est possible aussi de faire un INPUT AT en BASIC Sinclair:  
INPUT AT 5,0;AT 0,0;"Qui:";a\$  
Mais, encore, l'information à l'écran est effacée lorsque ENTER est enfoncée.

Pour remédier à cette situation, nous allons <sup>créer</sup> une petite routine en Langage Machine.

```
Assembleur: LD (IY+49),2
            RET
Decimal    : 253,54,49,2,201
```

Nous pouvons placer ces codes où nous voulons dans la mémoire, pour la démonstration, nous allons les mettre dans un REM. Tapez ce qui suit et RUN.

```
10 REM 12345
20 FOR i=26715 TO 26719
30 READ a:POKE i,a:NEXT i
40 DATA 253,54,49,2,201
```

Maintenant nous sommes prêt pour essayer notre nouveau INPUT AT. Effacez les lignes 20 à 50 et tapez celle-ci:

```
20 INPUT AT 10,0;AT 0,0;
   "Mois";m$;"Annee:";a
   'USR 26715
```

Le premier AT réserve 11 lignes à partir du bas de l'écran et le second va <sup>placer</sup> le INPUT sur la première de ses lignes.

### SCROLL PARTIEL...

La routine machine qui suit va "scroller" un certain nombre de ligne et autant de fois qu'on le veut.

Les codes qui suivent peuvent être à n'importe quel endroit en mémoire.

```
10 CLEAR 49999
20 FOR i=50000 TO 50020
30 READ a:POKE i,a:NEXT i
40 DATA 205,220,27,205,96,38,
        237,67,250,91,197,253,
        70,192,205,59,9,193,
        16,246,201
```

Pour utiliser cette routine, nous devons passer 2 arguments, si vous désassemblez cette routine, les 2 CALLs du début sont là à cette fin.

La syntaxe à respectée est la suivante:  
INPUT USR adresse, ligne, fois

Un exemple vaut 1000 mots, donc tapez le démo suivant et faites vos expériences.

```
10 CLS:PRINT "...Je ne bouge pas....."
20 INPUT AT 0,0;LINE a$
30 INPUT USR 50000,22,1
```

```
40 PRINT AT 21,0;a$
50 GOTO 20
```

### SAVE...

Si vous voulez remplacer le message "Start tape then press Enter" lors d'une sauvegarde de programme et vous n'avez qu'à faire:

```
POKE 26689,38
```

```
10 PRINT #0;"Embraye la cassette, mon vieux!"
```

```
20 POKE 26689,38:SAVE "x"
```

Pour ceux qui possède un ROM du SPECTRUM, voici un POKE qui effectue un SAVE sans attendre qu'une touche soit appuyée.

```
POKE 23736,181
```

### L O G I C I E L-----

Nom: TASWIDE (2068)

Type: Utilité

Cet utilitaire permet au TS 2068 d'afficher 64 caractères par ligne. Mais attention, ici il n'est pas question de l'utilisation du mythique Display Mode 3 de TIMEX.

Les caractères sont formés à partir d'une matrice 4 X 8, au lieu de 8 X 8 avec 32 car/ligne. Ceci entraîne une perte de lisibilité, surtout avec un téléviseur ordinaire, sur un moniteur monochrome on doit avouer que le résultat n'est pas si mal.

Ce logiciel est très simple d'utilisation. Après avoir chargé les codes, on met en branle le système avec un RAND

USR, puis pour écrire en 64 caractères un simple PRINT CHR\$ 3 suffit. Pour revenir en 32 caractères, on fait PRINT CHR\$ 2.

Ainsi pour listez en 64 car.:  
PRINT CHR\$ 3;: LIST

GLADSTONE vend ce logiciel pour la MODIQUE somme de 50 \$. Ce qui est cher, je pense. Si vous pouvez le trouver quelque part à un prix abordable, alors allez-y, il en vaut la peine.

Nom: WHITE LIGHTNING (SPEC)

Type: Langage

Si vous cherchez un FORTH, ne cherchez plus, vous l'avez trouvé. WL est un logiciel anglais, il vient en 2 cassettes. La première contient le langage FORTH, qui a été augmenté de 100 commandes permettant la manipulation de 255 SPRITES en même temps, et d'un programme pour faciliter la création de SPRITE. L'autre cassette contient d'incroyables démonstrations des possibilités de WL.

Si vous connaissez un peu le FORTH, vous savez que l'usage de l'éditeur n'est <sup>PAS</sup> chose aisée. Mais ici l'éditeur de la ROM est utilisé ce qui facilite bien les choses.

Je recommande très fortement ce logiciel. Mais attention, ne pensez <sup>PAS</sup> maîtriser ce langage en 1 semaine car c'est très complexe mais passionnant au possible.

Un endroit digne de confiance et très rapide pour vos logiciels SPECTRUM:  
SPEEDYSOFT, 87 Howard's Lane,  
London SW15 6NU, ENGLAND

Nom: COMPASS

Type: Utilite

Ce logiciel, écrit par John Coffey, est en fait 2 programmes complètement différents mais réunis en un seul. Il est composé d'un COMPILATEUR et d'un ASSEMBLEUR.

Le COMPILATEUR est un programme qui transforme un programme BASIC en code machine. Ceci est fait dans le but d'accélérer son déroulement. Ecrit entièrement en BASIC, le compilateur n'est pas d'une rapidité incroyable.

Donc vous écrivez votre programme BASIC et vous mettez en marche le compilateur. Mais ce n'est pas aussi simple. Car certaines commandes ne sont pas supportées. Des commandes aussi importantes que FOR...NEXT, DRAW, CIRCLE, INPUT et d'autres. En résumé, beaucoup de contrainte.

L'assembleur est lui aussi écrit en BASIC et est très lent. Vous écrivez vos mnémoniques dans des REMs.

La documentation pour le compilateur est complète et assez claire, tandis que celle de l'assembleur laisse à désirer. Si vous cherchez un Assembleur ne prenez pas celui-ci, optez plutôt pour ZEUS, produit par SOFTSYNC. Tant qu'au Compilateur il impose tellement de restrictions à l'écriture du programme BASIC, qu'il en perd tout son intérêt.

J.Coffey P.O. Box 448 Scottsburg In. 47170 USA, SOFTSYNC  
14 E 34th Street N.Y. 10016

## MATERIEL

Nom : ROMSWITCH

Si vous ne devez acheter qu'un périphérique pour votre 2068 alors c'est celui-ci. Il transforme votre 2068 en SPECTRUM à l'aide d'un dispositif d'une simplicité aberrante. Il est constitué d'un petit circuit intégré avec le ROM du SPECTRUM dessus, vous devez ouvrir votre 2068 pour installer votre ROM à côté. Vous avez alors 2 ordinateurs en 1. Un petit bouton aimanté est fourni pour permettre la mise en place du ROM du SPECTRUM ou du 2068.

Les instructions sont d'une limpidité à toute épreuve, aucune gaffe n'est possible. ROMSWITCH est fabriqué par RUSSELL ELECTRONICS, et vendu au environ de 50 \$ US. A Québec, vous pouvez le trouver chez GAGNON ELECTRONIQUE et si vous le désirez, ils peuvent l'installer pour vous.

RUSSELL ELECTRONIQUE  
RD#1 Box 539  
CENTRE HALL PA 16828  
USA

GAGNON ELECTRONIQUE  
390 Bl. Central Nord  
Duberger Québec P.Q. G1P 3P9  
Tel. 527-6103

---

### LE SAVIEZ-VOUS

Depuis le ZX 80, SINCLAIR a vendu plus de 5,000,000 d'ordinateurs dans le monde entier.

---

## PROGRAMME : R-HEADER

Le chargement d'un programme sur un TS 2068 se déroule habituellement en deux parties: 1-Le Header

2-Le programme/code/tableau

Ce programme est un utilitaire qui permet de lire le Header sans charger le programme. Les informations que contient un Header sont: 1-Le Type de programme (BASIC,Code,Tableau...) 2-Le Nom 3-Longueur avec et sans variable (BASIC) 4-Ligne Auto-Run (BASIC) 5-Origine des octets (Code)

### .LISTING 1 : R-HEADER

```
0> REM _____ R-HEADER 2068 GAG-085_____
5 BORDER 0: PAPER 0: BRIGHT 1: INK 5: CLS
10 CLEAR 31999
20 DATA 221,33,0,126,175,17,17,0,55,1,254,254,205,153,100,205,252,0,186,32,240,219,255,203,191,211
,255,175,211,244,201
40 FOR I=32000 TO 32030: READ A: POKE I,A: NEXT I
50 CLS : GO SUB 600: BEEP .01,10: PRINT AT 10,5: FLASH 1: INK 0: PAPER 6:"DEMARRER LA CASSETTE"
60 RANDOMIZE USR 32000
65 BEEP .01,45
69 CLS : GO SUB 600
70 LET ix=32256
80 LET type= PEEK ix
90 PRINT INVERSE 1>("Programme:" AND type=0)+("Tableau numerique:" AND type=1)+("Tablea
u de caractere:" AND type=2)+("Bloc d'octets:" AND type=3);
100 PRINT INVERSE 0;" "; FOR n=ix+1 TO ix+10: PRINT CHR$ PEEK n; NEXT n
110 PRINT INVERSE 1'"Longueur "+("Prog. + Variable:" AND type=0)+("Code:" AND type)
;
120 PRINT INVERSE 0;" "; PEEK (ix+11)+256* PEEK (ix+12)
130 IF type=1 OR type=2 THEN PRINT INVERSE 1;"Variable:"; PRINT INVERSE 0;" "; CHR$ ( PEEK
(ix+14)-32-64*( PEEK (ix+14)>192))+("$" AND type=2): GO TO 175
140 PRINT ": PRINT INVERSE 1;("Auto-Run Ligne:" AND type=0)+("Origine des octets:" AND typ
e=3);
150 PRINT INVERSE 0;" "; PEEK (ix+13)+256* PEEK (ix+14): POKE ix+13,0: POKE ix+14,0
155 IF type=3 THEN GO TO 180
160 PRINT INVERSE 1'"Longueur du Programme:" AND type=0);
170 PRINT INVERSE 0;" "; PEEK (ix+15)+256* PEEK (ix+16)
175 PRINT ''' OVER 1;"....."
180 PRINT '' OVER 1;" S: STOP ENTER: Autre lecture": PAUSE 0: IF INKEY$ ="s" OR INKEY$ ="S" THEN
STOP
190 RUN
600 PLOT 0,0: DRAW 255,0: DRAW 0,175: DRAW -255,0: DRAW 0,-175: PRINT INVERSE 1; AT 0,9;"R-HEADER
2068"; AT 1,9;" GAGNON-085 '''
610 RETURN
9991 SAVE "R-HEADER" LINE 1: VERIFY "R-HEADER"
```

## GRAPHISME

```
10 REM SPIRALES
20 INPUT x,y :REM x,y=<1
30 INPUT p : REM cotes
40 PLOT 128,88
50 FOR a=0 TO 500*PI STEP
    PI/p
60 DRAW x*a*SIN a,y*a*COS a
70 NEXT a

10 REM SINUS
20 FOR a=.5 TO 12.9 STEP .1
30 LET b=a*10:LET c=87+40*SIN a
40 PLOT b,c :DRAW b,c/5
50 NEXT a

10 REM Ce programme fait apparaitre
    un message doucement...
15 LET z$="Ceci est une demonstration..."
20 FOR j=1 TO LEN z$
25 LET ch=8*(CODE z$(j)-32)+15616
30 LET zz=20200+j-1+15*(LEN z$/2)
40 FOR i=1 TO 6: BEEP .001,60
50 POKE zz+i*256,PEEK (ch+i)
60 NEXT i: NEXT j: BEEP .01,-10: PAUSE 0
70 LET z$="Avez-vous bien compris,hein!?! "
80 GOTO 20

10 REM POLYGONE
11 PAPER 0: BORDER 0: INK 0: CLS
12 INPUT "Combien de cotes? ";n
13 CLS : IF n <= 0 THEN GOTO 12
14 SOUND 8,56;8,10;9,10;10,10
15 DIM x(n): DIM y(n)
16 PRINT AT 0,8;"Figure a ";n;" cotes"
17 LET a=2*PI/n: LET cx=120 : LET cy=90: LET s=70
20 FOR i=1 TO n: LET t=a*(i-1)
21 LET x(i)=cx+s*COS t: LET y(i)=cy+s*SIN t: NEXT i
30 FOR i=1 TO n-1: FOR j=1+1 TO n
31 SOUND 0,5*i+j;2,3*j;4,5*j+i
32 PLOT x(i),y(i): DRAW x(j)-x(i),y(j)-y(i)
33 NEXT j: NEXT i: SOUND 8,0;9,0;10,0
34 PRINT #0;"Une touche...": PAUSE 0: GOTO 12

10 REM Chapeau Haute-Resolution (Soyez patient)
    extrait de TS HORIZONS (juin 85)
20 PLOT 0,0: DRAW 0,175: DRAW 255,0: DRAW -175,0: DRAW -255,0
30 LET xs=.78: LET ys=.85: LET p=128: LET q=95: LET xp=144
40 LET xr=1.5*PI: LET yp=56: LET yr=1: LET zp=64: LET xf=xr/xp
50 LET yf=yp/yr: LET zf=xr/zp: LET xp=xp/zp
60 FOR k=-zp TO zp: LET zt=k*xq
70 LET x1=INT (.5+SQR (xp*xp-zt*zt))
80 FOR t=-x1 TO x1:LET xt=SQR (t*t+zt*zt)*xf:LET x=(xs*(t+k))+p
90 LET yy=(SIN (xt)+.4*SIN (3*xt))*yf: LET y=(ys*(yy-k))+q
95 PLOT x,y: NEXT t: NEXT k
```



Le programme MASTERMIND s'adresse a votre esprit logique.  
 Vous devez trouver une combinaison de 4 couleurs que l'ordinateur  
 a choisie.

Ce programme contient quelques routines de traitement de  
 chaine de caractères qui peuvent s'adapter a vos propres  
 programmes assez facilement.

```

1 REM _____MASTERMIND GAGNON-o 1985
60 DIM a(4): DIM b(4): DIM c(4): PAPER 0: BORDER 0: INK 7: CLS
75 REM _____COMBINAISON
80 FOR j=1 TO 4: LET a(j)= INT ( RND *6)+1: NEXT j
85 REM _____L'ECRAN
86 LET G#= CHR$ 143: REM Caractere Graphique #143
100 PRINT ' : INK 0: FOR i=0 TO 9: BEEP .0005,57: BRIGHT 0: PRINT TAB 2: BRIGHT 1;G#;G#;G#;G#;G#
110 PRINT TAB 2: BRIGHT 1;G#: BRIGHT 0: " "; BRIGHT 1;G#
120 BEEP .0001,60: NEXT i
125 PRINT TAB 2: BRIGHT 1;G#;G#;G#;G#;G#;G#
126 BRIGHT 0
130 PRINT #0: BRIGHT 1;"Bon,bien place [*].""Bon,mal place [/]."
150 INK 6: FOR x=158 TO 7 STEP -16: BEEP .0001,60: PLOT 78,x: DRAW 35,0: BEEP .0005,57: DRAW 0,10:
DRAW -35,0: DRAW 0,-10: NEXT x
160 INK 7: GO SUB 1000
165 REM _____L'ENTREE AU CLAVIER
190 LET x=1: PRINT AT x,1;">": AT x,8;"<": FOR y=1 TO 4: PAUSE 0
200 IF INKEY$ >"6" OR INKEY$ <"1" THEN GO TO 460
210 PRINT AT x,y+2: INK VAL INKEY$ :G#;
220 LET b(y)= VAL INKEY$ : IF y=4 THEN BEEP .02,20
240 NEXT y: GO SUB 310
260 PRINT AT x,1;" ": AT x,8;" ": LET x=x+2: IF x>20 THEN GO TO 510
270 GO TO 190
300 REM _____VERIFICATION
310 LET p=10: LET e=0: LET d=0: LET a$="****": LET b$="////"
330 FOR i=1 TO 4: LET c(i)=a(i): NEXT i
340 FOR j=1 TO 4: IF c(j)=b(j) THEN LET e=e+1: LET b(j)=0: LET c(j)=0: NEXT j
360 FOR j=1 TO 4: FOR k=1 TO 4
370 IF b(j)=c(k) AND b(j)>0 THEN LET d=d+1: LET b(j)=0: LET c(k)=0
380 NEXT k: NEXT j
400 PRINT AT x,p;a$( TO e);b$( TO d)
410 IF e=4 THEN FOR y=0 TO 4: FOR i=0 TO 6: PAUSE 1: PRINT AT 11,22: INK i;"BRAVO!": BEEP .0001,5
5: NEXT i: NEXT y: PRINT AT 11,22: INK 7;"BRAVO!": BEEP .01,30: GO TO 510
420 RETURN
450 REM _____ENTREE INCORRECT/EFFACEMENT
460 PRINT AT x,3;" ": BEEP .1,-10: GO TO 190
500 REM _____LE CODE EST ...
510 PLOT 155,28: DRAW 95,0: DRAW 0,40: DRAW -95,0: DRAW 0,-40: PRINT AT 14,20;"Le code est"
520 FOR y=1 TO 4: BEEP .001,30+(5*y): PRINT AT 16,y+22: INK a(y);G#: NEXT y
550 REM _____UNE AUTRE PARTIE?
560 PRINT AT 20,17;"Une Autre(o/n)?": PAUSE 0: IF INKEY$ ="o" THEN RUN
570 IF PEEK 23556= CODE "N" THEN STOP
580 GO TO 560
999 REM _____TITRE
1000 LET q$="*MASTERMIND*": LET c=13
1010 FOR i=1 TO 12: LET a$=q$(i)

```

(suite au verso)

## SOURCE D'INFORMATION

### LIVRES, LOGICIELS, PERIPHERIQUES...

GAGNON ELECTRONIQUE	390 BL. CENTRAL NORD DUBERGER QUEBEC G1P 3P9 Tel. 527-6103
GLADSTONE ELECTRONICS	1736 AVENUE RD. TORONTO ONTARIO M5M 3Y7
CURRY COMPUTER	5344 W. BANFF GLENDALE AZ 85306 USA
RESEARCH SERV. LABS	P.O BOX 19124 OKLAHOMA CITY OK 73144 USA
SPEEDY SOFT	87 HOWARD'S LANE LONDON SW15 6NU ENGLAND

### MAGAZINES...

ZX COMPUTING	PO BOX 35 WOLSEY HOUSE, WOLSEY ROAD HEMEL HEMPSTEAD, HERTS HP2 4SS ENGLAND
SINCLAIR USER	PRIORY COURT, 30-32 FARRINGTON LANE LONDON EC1R 3AU ENGLAND
YOUR COMPUTER	OAKFIELD HOUSE PERRYMOUNT ROAD, HAYWARDS HEATH SUSSEX RH16 3DH ENGLAND
TS HORIZONS	2002 SUMMIT STREET PORTSMOUTH OH 45662 USA
SUM	3224 NW 30 AVENUE GAINESVILLE FL 32605 USA

---

### PROCHAIN NUMERO

Programme: Générateur de caracteres  
Utilisation du Mode Ultra-Haute Résolution du 2068  
STICK/SOUND pour le SPECTRUM 2068  
la situation du QL et bien d'autres...  
(le prochain numéro devrait sortir fin août-début septembre 85)

---

Si vous avez des commentaires/suggestions/questions, vous pouvez  
me les faire parvenir à: Réal Gagnon, 544 Hermine #2, Québec, P.Q.  
G1N 2G6

• Suite de la  
page précédente  
(MASTERMIND)

```
1020 FOR y=0 TO c: BEEP .00005,64: PRINT AT 0,30-y: INK 6: a$; INK 7; " ": NEXT y
1030 LET c=c-1: NEXT i
1061 LET q$="c GAGNON-o1984 "
1062 FOR i=1 TO 14: PRINT AT 1,30-i;q$( TO i); " ": NEXT i
1063 BEEP .05,-25: LET c$="COULEURS": LET p$="POSSIBLES"
1064 LET u$="172737405060"
1065 FOR i=0 TO 1: FOR y=19 TO 6 STEP -1
1066 IF i AND y=6 THEN GO TO 1071
1070 PRINT AT y,(19 AND NOT i)+(20 AND i):(c$ AND NOT i)+(p$ AND i): AT y+1,19;"
1071 BEEP .00005,64: NEXT y: NEXT i
1072 PRINT AT 9,19;: FOR i=1 TO 11 STEP 2: PRINT PAPER VAL u$(i); INK VAL u$(i+1);u$(i); PAPER 0
;" ": BEEP .001,i+50: NEXT i
1090 BEEP .02,-10: RETURN
```

## LES ERREURS.....

Nous vous en donnons les corrections avec toutes nos excuses.

### Poygone, page 7

```
14 SOUND 7,56;8,10;9,10;10,10
30 FOR i=1 TO n-1: FOR j=i+1 TO n
```

### Chapeau, page 7

```
20 PLOT 0,0: DRAW 0,175: DRAW 255,0: DRAW 0,-175: DRAW -255,0
50 LET yf=yp/yr: LET zf=xr/zp: LET xq=xp/zp
```

-----